

MATH 2270-1
MAPLE PROJECT 2b - Inner product spaces
 October 23, 2005

We will explore two interesting inner product spaces. Refer to section 5.5 of the text. This part of your project replaces homework problems from that section.

A good dot product and interval for orthogonal polynomials:

The first inner product and interval we mention are used in numerical analysis algorithms. We'll call the inner product "dot1". It is the inner product of example 7, page 237.

```
[ > restart:with(plots):
Warning, the name changecoords has been redefined
> dot1:=(f,g)->int(f(t)*g(t),t=-1..1);
    #our first inner product, for the
    #interval -1<t<1
```

$$dot1 := (f, g) \rightarrow \int_{-1}^1 f(t) g(t) dt$$

```
[ > dot1(t->t,t->1);
    #the dot product of f(t)=t with g(t)=1
    #you should get zero. Meaning what?

0
```

Meaning you just showed that $f(t)=t$ and $g(t)=1$ are orthogonal for the dot1 inner product! Does this surprise you? (Think of the graph of these two functions for $-1 < t < 1$ and think about what the integral of $f(t)g(t)$ is measuring.) In fact, once the inner product is defined, you can define magnitude, distance, angle between vectors, and you can even do projection problems. This is discussed in detail in section 5.5, and we'll add to the discussion in class on Tuesday.

```
[ > mag1:=f->sqrtdot1(f,f):
    # the magnitude of a vector

[ > dist1:=(f,g)->mag1(f-g):
    # the ''distance'' between two functions
[ > cosangle1:=(f,g)->(dot1(f,g)/(mag1(f)*mag1(g))):
    #computes the "cos of the angle" between functions
[ > fangle1:=(f,g)->evalf(arccos(cosangle1(f,g))):
    #computes angle between functions
```

For example we may use Gram-Schmidt to find an orthonormal basis for the polynomial subspace $P_2 = \text{span}\{1, t, t^2\}$, for our inner product.

```
[ > P0:=t->1;
    P1:=t->t;
    P2:=t->t^2;
    #our usual "natural" basis

    P0 := 1
```

```


                                 $P1 := t \rightarrow t$ 
                                 $P2 := t \rightarrow t^2$ 
> W0:=P0/mag1(P0):    #first orthonormal vector
  Z1:=P1-dot1(P1,W0)*W0:
    #P1 was orthogonal to P0, so don't really
    #need the usual projection formula, but
    #here it is anyway.
  W1:=Z1/mag1(Z1):
    #second orthonormal vector
  Z2:=P2-dot1(P2,W0)*W0-dot1(P2,W1)*W1:
  W2:=Z2/mag1(Z2):
    #third orthonormal vector
> W0(t);W1(t);W2(t);
    #the orthonormal polynomials we get
                                
$$\frac{\sqrt{2}}{2}$$

                                
$$\frac{t\sqrt{6}}{2}$$

                                
$$\frac{3\left(t^2 - \frac{1}{3}\right)\sqrt{10}}{4}$$

> plot({W0(t),W1(t),W2(t)},t=-1..1, title='orthonormal basis
  for P2, using dot1');

```



Projection:

Now that we have an orthonormal basis for P_2 we can do projection problems. We will try to find the closest degree 2 polynomial to $f(t)=\exp(t)$, using our inner product dot1 to measure distance:


```
[ > proj2:=f->evalf(dot1(f,W0)*W0+dot1(f,W1)*W1+dot1(f,W2)*W2);  
    #this is the usual projection formula, but we use evalf to  
    #get decimals rather than messy algebraic numbers.  
    proj2:=f->evalf(dot1(f,W0)W0+dot1(f,W1)W1+dot1(f,W2)W2)  
[ > f:=t->exp(t);  
                                     f:=exp  
[ > g:=t->proj2(f)(t);  
                                     g:= 0.9962940173 + 1.103638324 P1 + 0.5367215271 P2
```

This polynomial g is "closer" to $\exp(t)$ than the usual Taylor polynomial $p(t)=1+t+t^2/2$, when we use the distance which we get from dot1 . We can compare distances:

```
[ > p:=t->1+t+t^2/2;  
                                     p:=t -> 1 + t + 1/2 t^2  
[ > evalf(dist1(f,g));  
    evalf(dist1(f,p));  
                                     0.03795488813  
                                     0.09479756619  
[
```

So, at least for our distance, the function g does about three times as well as the Taylor polynomial. You can also see this geometrically, by plotting the three graphs on the interval from -1 to 1:

```
[ > with(plots):  
[ > actual:=plot(f(t),t=-1..1,color=red):  
    best:=plot(g(t),t=-1..1,color=black):  
    tayl:=plot(p(t),t=-1..1,color=blue):  
    display({actual,best,tayl});
```



The orthogonal polynomials which you have been constructing are called the Legendre polynomials. Maple knows about them. To see the first few you can load the ‘orthopoly’ package. By the way, if you define different weighted inner products you get different (famous to experts) orthogonal polynomial families, you can read about some of them on the help windows, starting at ‘orthopoly’. Orthogonal polynomials are used in approximation problems, as you might expect.

```
[ > with(orthopoly):
> P(0,x);P(1,x);P(2,x);P(3,x);P(4,x);P(5,x);
#these should look familiar, after what you just did!
#they haven't been normalized, though.
```

$$\begin{aligned}
 &1 \\
 &x \\
 &-\frac{1}{2} + \frac{3x^2}{2} \\
 &\frac{5}{2}x^3 - \frac{3}{2}x \\
 &\frac{3}{8} + \frac{35}{8}x^4 - \frac{15}{4}x^2 \\
 &\frac{63}{8}x^5 - \frac{35}{4}x^3 + \frac{15}{8}x
 \end{aligned}$$

Fourier Series:

Here's an interval and dot product which makes the usual trig functions into an orthonormal family! See page 238 of the text!

```
> dot2:=(f,g)->1/Pi*int(f(t)*g(t),t=-Pi..Pi);
```

$$dot2 := (f, g) \rightarrow \frac{1}{\pi} \int_{-\pi}^{\pi} f(t) g(t) dt$$

For example:

```
> f:=t->sin(5*t);  
g:=t->cos(3*t);  
dot2(f,f);  
dot2(g,g);  
dot2(f,g);
```

$$f := t \rightarrow \sin(5 t)$$

$$g := t \rightarrow \cos(3 t)$$

1

1

0

We add the constant function to our collection, but it will need to be normalized:

```
> dot2(1,1);  
dot2(f,1);  
dot2(g,1);  
#the constant function has norm squared  
#equal to 2, however. But it is orthogonal  
#to all cos(kt),sin(kt), k a natural number
```

2

0

0

Thus, for any n, the family $\{1/\sqrt{2}, \cos(t), \cos(2t), \dots, \cos(nt), \sin(t), \sin(2t), \dots, \sin(nt)\}$ is an orthonormal basis of a $2n+1$ dimensional subspace of functions. So it is easy to project onto this subspace using our usual projection formulas. It is an amazing fact that if $f(t)$ is any piecewise continuous function on the interval $-\pi \leq t \leq \pi$, then as n approaches infinity the distances between these projections and f converges to zero. Here's an example, see page 241:

```
> f:=t->t;  
#the function we shall decompose into trigonometric pieces
```

```
> a0:=int(f(t),t=-Pi..Pi);  
#could you have predicted the answer?
```

$$a0 := 0$$

```
> #now get projection coefficients  
#AKA Fourier coefficients  
for i from 1 to 10 do  
g1:=t->cos(i*t):
```

```

g2:=t->sin(i*t):
a[i]:=dot2(g1,f):
b[i]:=dot2(g2,f):
od:
>
> fsum:=t-> a0/(2*Pi)+ sum(a[j]*cos(j*t),j=1..10) +
    sum(b[j]*sin(j*t),j=1..10);

```

$$fsum := t \rightarrow \frac{a_0}{2\pi} + \left(\sum_{j=1}^{10} a_j \cos(jt) \right) + \left(\sum_{j=1}^{10} b_j \sin(jt) \right)$$

```

> fsum(t);
    #why are there no cosine terms?


```

$$2 \sin(t) - \sin(2t) + \frac{2}{3} \sin(3t) - \frac{1}{2} \sin(4t) + \frac{2}{5} \sin(5t) - \frac{1}{3} \sin(6t) + \frac{2}{7} \sin(7t) - \frac{1}{4} \sin(8t) + \frac{2}{9} \sin(9t) - \frac{1}{5} \sin(10t)$$

```

> plot1:=plot(fsum(t),t=-Pi..Pi,color=black):
plot2:=plot(t,t=-Pi..Pi,color=red):
display({plot1,plot2});

```



```

[ >

```

There are problems for you to do! Download the file 2270proj2b.mws from the project 2 web page.